Mathematics 1110H (Section A) – Calculus I: Limits, Derivatives, and Integrals
Trent University, Fall 2024

Solutions to Assignment #3
Differential Equations

If you haven't already seen it, look up SageMath's `desolve` command, which is used to solve differential equations.

1. Consider the differential equation $x\dfrac{dy}{dx} = y$.

a. Use SageMath to find a general solution to this differential equation. *[1]*

b. Use SageMath to find a solution to this differential equation satisfying the initial condition $y = 1$ when $x = 1$. *[1]*

c. Verify – by hand! – that the solution you obtained in part **b** satisfies the given differential equation with the given initial conditions. *[0.5]*

Solutions. **a.** Here it is:

```
[1]: # SageMath solutions for MATH 1110H-A Assignment #3

     y = function('y')(x) # so we can differentiate y
     z = diff(y,x)        # so we don't have to keep typing diff(y,x)

     # 1a
     desolve( x*z == y, y )

[1]: _C*x
```

That is, the general solution to $x\dfrac{dy}{dx} = y$ is $y = Cx$, where C is a constant. □

b. Here is the version with $y = 1$ when $x = 1$:

```
[2]: # 1b
     desolve( x*z == y, y, ics=(1,1) )

[2]: x
```

That is, the particular solution to $x\dfrac{dy}{dx} = y$ which has $y = 1$ when $x = 1$ is $y = x$. This corresponds, of course, to setting $C = 1$ in the general solution. □

c. If $y = x$, then $\dfrac{dy}{dx} = \dfrac{d}{dx}x = 1$, so

$$x\frac{dy}{dx} = x \cdot 1 = x = y$$

and $y = x = 0$ when $x = 0$, so the solution $y = x$ indeed satisfies the given differential equation with the given initial condition. ■

2. Consider the differential equation $\dfrac{dy}{dx} = xy$.

- **a.** Use SageMath to find a general solution to this differential equation. *[1]*
- **b.** Use SageMath to find a solution to this differential equation satisfying the initial condition $y = 1$ when $x = 0$. *[1]*
- **c.** Verify – by hand! – that the solution you obtained in part **b** satisfies the given differential equation with the given initial conditions. *[0.5]*

SOLUTIONS. **a.** Here it is:

```
[3]: # 2a
     desolve( z == x*y, y )
```

```
[3]: _C*e^(1/2*x^2)
```

That is, the general solution to $\dfrac{dy}{dx} = xy$ is $y = Ce^{x^2/2}$, where C is a constant. □

b. Here is the version with $y = 1$ when $x = 0$:

```
[4]: # 2b
     desolve( z == x*y, y, ics=(0,1) )
```

```
[4]: e^(1/2*x^2)
```

That is, the particular solution to $\dfrac{dy}{dx} = xy$ which has $y = 1$ when $x = 0$ is $y = e^{x^2/2}$. This corresponds, of course, to setting $C = 1$ in the general solution. □

c. If $y = x$, then $\dfrac{dy}{dx} = \dfrac{d}{dx}e^{x^2/2} = e^{x^2/2}\dfrac{d}{dx}\left(\dfrac{x^2}{2}\right) = e^{x^2/2}\dfrac{2x}{2} = xe^{x^2/2}$. It follows that

$$\frac{dy}{dx} = xe^{x^2/2} = xy$$

and $y = e^{0^2/2} = e^0 = 1$ when $x = 0$, so the solution $y = e^{x^2/2}$ indeed satisfies the given differential equation with the given initial condition. ■

3. Consider the differential equation $\dfrac{dy}{dx} = y^2 + 1$.

- **a.** Use SageMath to help find a general solution to this differential equation. *[1]*
- **b.** Use SageMath to help find a solution to this differential equation satisfying the initial condition $y = 0$ when $x = 0$. *[1]*
- **c.** Verify – by hand! – that the solution you obtained in part **b** satisfies the given differential equation with the given initial conditions. *[0.5]*

SOLUTIONS. **a.** Here it is:

```
[5]: # 3a
     desolve( z == y^2 + 1, y )
```

```
[5]: arctan(y(x)) == _C + x
```

This is a bit less than what we want, since y has not actually been solved for.

One can remedy this by hand, $\arctan(y) = x + C \iff y = \tan(x + C)$, using the fact that tan and arctan are each other's inverese functions, or by handing off the problem to SageMath's regular `solve` command:

```
[6]: solve( arctan(y(x)) == _C + x, y(x) )
```

```
 ␣
↪---------------------------------------------------------------------------

    NameError                                 Traceback (most recent call␣
↪last)

    <ipython-input-6-cb7bddc3622a> in <module>()
----> 1 solve( arctan(y(x)) == _C + x, y(x) )


    NameError: name '_C' is not defined
```

Oops! It seems SageMath needs the unknown constant to be pinned down before `solve` will do the job. Good thing we can also do this by hand ...

Thus the general solution to $\frac{dy}{dx} = y^2 + 1$ is $y = \tan(x + C)$, where C is a constant. □

b. Here is the version with $y = 0$ when $x = 0$:

```
[7]: # 3b
     desolve( z == y^2 + 1, y, ics=(0,0) )
```

```
[7]: arctan(y(x)) == x
```

As with the general case, y is isn't actually solved for ... Crossing our fingers, we try using the `solve` command again, since this time there are no unknown constants for it to trip over:

```
[8]: solve( arctan(y(x)) == x, y(x) )
```

```
[8]: [y(x) == tan(x)]
```

That is, the particular solution to $\frac{dy}{dx} = y^2 + 1$ which has $y = 0$ when $x = 0$ is $y = \tan(x)$. This corresponds, of course, to setting $C = 0$ in the general solution. □

c. If $y = \tan(x)$, then $\frac{dy}{dx} = \frac{d}{dx}\tan(x) = \sec^2(x) = \tan^2(x) + 1 = y^2 + 1$, and we also have $y = \tan(0) = 0$ when $x = 0$, so the solution $y = \tan(x)$ indeed satisfies the given differential equation with the given initial condition. ■

4. Use SageMath to help find all the general solutions to the differential equation

$$\left(\frac{dy}{dx}\right)^2 + (x+y)\frac{dy}{dx} + xy = 0.$$

Explain why the solutions you found work. *[2.5]*

Hint: A little preliminary algebra can make this much easier.

SOLUTION 1. *Ignoring the hint.* We plug the given differential equation into `desolve` and see what happens:

```
[9]: # 4
desolve( z^2 + (x+y)*z + x*y == 0, y )
```

```
␣
↪---------------------------------------------------------------------------

    NotImplementedError                       Traceback (most recent call␣
↪last)

    <ipython-input-9-f87508f8d9fd> in <module>()
----> 1 desolve( z**Integer(2) + (x+y)*z + x*y == Integer(0), y )


    /opt/conda/envs/sage/lib/python3.7/site-packages/sage/calculus/desolvers.
↪py in desolve(de, dvar, ics, ivar, show_method, contrib_ode, algorithm)
    595                 raise NotImplementedError("Maxima was unable to␣
↪solve this ODE.")
    596         else:
--> 597             raise NotImplementedError("Maxima was unable to solve␣
↪this ODE. Consider to set option contrib_ode to True.")
    598
    599     if show_method:


    NotImplementedError: Maxima was unable to solve this ODE. Consider to␣
↪set option contrib_ode to True.
```

Implementing the suggestion at the end of the error message above, we try again:

```
[10]: desolve( z^2 + (x+y)*z + x*y == 0, y, contrib_ode=True )
```

```
[10]: [y(x) == -1/2*x^2 + _C, y(x) == _C*e^(-x)]
```

(Yay! A useful error message! :-)

Thus, this differential equation has two possible solutions, $y = -\dfrac{x^2}{2} + C$ and $y = Ce^{-x}$, where C is a constant either way. △

Solution 2. *Using the hint.* Observe that

$$\left(\frac{dy}{dx}\right)^2 + (x+y)\frac{dy}{dx} + xy = \left(\frac{dy}{dx} + x\right)\left(\frac{dy}{dx} + y\right).$$

It follows that $\left(\frac{dy}{dx}\right)^2 + (x+y)\frac{dy}{dx} + xy = 0$ exactly when $\frac{dy}{dx} + x = 0$ or $\frac{dy}{dx} + y = 0$. The latter two differential equations are pretty easy to solve by hand – give them a try! – or by using SageMath:

```
[11]: desolve( z + x == 0, y )

[11]: -1/2*x^2 + _C

[12]: desolve( z + y == 0, y )

[12]: _C*e^(-x)
```

Thus, this differential equation has two possible solutions, $y = -\frac{x^2}{2} + C$ and $y = Ce^{-x}$, where C is a constant either way. △

For both solutions. We still need to explain why these solutions work, which we can do by simply plugging them into the differential equation and see what happens. Observe that $\frac{d}{dx}\left(-\frac{x^2}{2} + C\right) = -\frac{2x}{2} + 0 = -x$ and $\frac{d}{dx}\left(Ce^{-x}\right) = Ce^{-x}\frac{d}{dx}(-x) = Ce^{-x}(-1) = -Ce^{-x}$. Plugging the first solution into the original differential equation, so $y = -\frac{x^2}{2} + C$ and $\frac{dy}{dx} = -x$, gives

$$\begin{aligned}\left(\frac{dy}{dx}\right)^2 + (x+y)\frac{dy}{dx} + xy &= (-x)^2 + \left(x - \frac{x^2}{2} + C\right)(-x) + x\left(-\frac{x^2}{2} + C\right)\\ &= x^2 - x^2 + \frac{x^3}{2} - Cx - \frac{x^3}{2} + Cx = 0,\end{aligned}$$

so the first solution does satsfy the given differential equation. Similarly, plugging the second solution into the original differential equation, so $y = Ce^{-x}$ and $\frac{dy}{dx} = -Ce^{-x}$, gives

$$\begin{aligned}\left(\frac{dy}{dx}\right)^2 + (x+y)\frac{dy}{dx} + xy &= \left(-Ce^{-x}\right)^2 + \left(x + Ce^{-x}\right)\left(-Ce^{-x}\right) + xCe^{-x}\\ &= C^2e^{-2x} - xCe^{-x} - C^2e^{-2x} + xCe^{-x} = 0,\end{aligned}$$

so the second solution also satisfies the given differential equation. ■